MACROS 1

```
/************************************************
************************************************
   We will start with creating a simple MACRO
   which will provide a histogram and boxplot
   for one quantitative variable

   Before creating the macro we must start by
   Testing the Basic Code without Macro Variables
************************************************
************************************************/

proc sgplot data=AMY.fghm113;
histogram age;
density age / type=normal;
density age / type=kernel;
run;

proc sgplot data=AMY.fghm113;
vbox age;
run;
```

```
/**************************************************
**************************************************
 Macro Creation

 Here we must determine which specific items should be
 allowed to change in future use of this code

 We will allow the dataset and variable name to change

 We will also add one option allowing user to choose
 horizontal or vertical boxplots
**************************************************
**************************************************/

%macro graphs(graphsdata, graphsvar, boxtype=VBOX);
proc sgplot data=&graphsdata;
histogram &graphsvar;
density &graphsvar / type=normal;
density &graphsvar / type=kernel;
run;

proc sgplot data=&graphsdata;
&boxtype &graphsvar;
run;
%mend;

/**************************************************
 (graphsdata, graphsvar, boxtype=VBOX) tells sas these will
 be macro variables defined when we run this macro and are
 called parameters

 The first two are positional parameters which should be
 specified in order without labels.

 Since we defined boxtype to be VBOX by default if we
 do not specify this macro variable it will use VBOX.

 When we specify boxtype, we must specify the parameter name
 such as boxtype = VBOX
**************************************************/
```

```
/*******************************************************
*******************************************************
 Create PDF file of output of running graphs macro with
 various settings
*******************************************************
*******************************************************/

ods pdf file = "H:\_SAS\MACROS1.pdf" notoc startpage=never;

ods graphics / width=5in height=4in;

Ods escapechar='^';
Ods PDF text="^2n";
Ods PDF text="^S={font=('Times Roman',12pt,Bold)}Simple Macro using
AGE";

/*******************************************************
 Run macro with default boxtype on AGE in FGHM113 data
*******************************************************/

%graphs(AMY.fghm113, age)

ods pdf startpage = now;

Ods PDF text="^2n";
Ods PDF text="^S={font=('Times Roman',12pt,Bold)}Simple Macro using
TOTCHOL";

/*******************************************************
 Run macro with boxtype = HBOX on TOTCHOL in FGHM113 data
*******************************************************/

%graphs(AMY.fghm113, totchol, boxtype=hbox)

Ods PDF text="^S={font=('Times Roman',12pt,Bold)}Simple Macro using
WTLBS in NHANES data";

/*******************************************************
 Run macro with boxtype = HBOX on WTLBS in NHG data
*******************************************************/

%graphs(TEMP.nhg, wtlbs,  boxtype=hbox)

ods pdf close;
```

```
/*********************************************
*********************************************
 Illustrate potential errors in executing MACRO
*********************************************
*********************************************/

/* What happens if we incorrectly specify the boxtype? */
%graphs(TEMP.nhg, wtlbs, boxtype=hbar)
%graphs(TEMP.nhg, wtlbs, boxtype=hplot)

/* What happens if we incorrectly specify the dataset or variable
name? */
%graphs(TEMP.spelling, wtlbs)
%graphs(TEMP.nhg, wtpnds)
```

MACROS 2

```
/*********************************************************
**********************************************************
 Now we add looping for numerous Quantitative variables

 We add a parameter called nvar and an s to the end of
 graphsvar indicating this is a LIST of variable names

 The %let statement uses the %scan function to choose the
 i-th word (variable name) in the LIST graphsvars
**********************************************************
*********************************************************/

%macro graphsloop(graphsdata, graphsvars, nvar, boxtype=VBOX);

%do i = 1 %to &nvar;
     %let currentvar = %scan(&graphsvars, &i);
     proc sgplot data=&graphsdata;
     histogram &currentvar;
     density &currentvar / type=normal;
     density &currentvar / type=kernel;
     run;

     proc sgplot data=&graphsdata;
     &boxtype &currentvar;
     run;
%end;

%mend;
```

```
/*****************************************************
*****************************************************
 Run looping macro on all 9 quantitative variables in
 the FGHM113 dataset
*****************************************************
*****************************************************/

ods pdf file = "H:\_SAS\MACRO2A.pdf" notoc startpage=never;

ods graphics / width=5in height=4in;

Ods escapechar='^';
Ods PDF text="^2n";
Ods PDF text="^S={font=('Times Roman',12pt,Bold)}Looping Macro for
all 9 Quantitative Variables in FGHM113";

%graphsloop(AMY.fghm113,
                age totchol bmi hdlc ldlc glucose heartrte diabp
sysbp, 9)

ods pdf close;



/*****************************************************
*****************************************************
 Using MLOGIC and MPRINT
*****************************************************
*****************************************************/

/* Using MLOGIC to see macro processing in LOG file */
option mlogic;

%graphsloop(AMY.fghm113,
                age totchol, 2)

/* Adding MPRINT to see all code processed */
option mprint;
%graphsloop(AMY.fghm113,
                bmi hdlc ldlc, 3)

/* Turn off MLOGIC and MPRINT */
option nomlogic nomprint;
```

```
/*********************************************
**********************************************
 Trick for obtaining list of variables quickly
**********************************************
**********************************************/

/* print first few observations of data */
proc print data=temp.nhg (obs=2);
run;

/*************************************************************
 Copy line of variable names and paste in SAS code comment
 AGE SEX RACE WTLBS HTIN SBP DBP TCP SMOKE HBP BMI LogSBP

 Edit as needed - here we want the quantitative variables

 AGE WTLBS HTIN SBP DBP TCP BMI LogSBP
*************************************************************/


/*************************************************************
**************************************************************
 Run macro on NHG data which has 8 quantitative variables
**************************************************************
**************************************************************/

ods pdf file = "H:\_SAS\MACRO2B.pdf" notoc startpage=never;

ods graphics / width=5in height=4in;

Ods escapechar='^';
Ods PDF text="^2n";
Ods PDF text="^S={font=('Times Roman',12pt,Bold)}Looping Macro for
all 8 Quantitative Variables in NHG with BOXTYPE=HBOX";

%graphsloop(temp.nhg,
                AGE WTLBS HTIN SBP DBP TCP BMI LogSBP, 8,
boxtype=hbox)

ods pdf close;
```

MACROS 3

```
/*************************************************
 This code creates a looping macro which finds the
 p-value and reports whether or not the association
 is significant
*************************************************/
Options nodate nonumber;

/*  The dataset contains permanently stored formats,
    you must submit the following code or use the
    option NOFMTERR in the options statement above
    to be able to use the WHAS500 SAS dataset       */

proc format;
value YesNoFmt 1='Yes'
               0='No';
value Sex01Ft  0='Male'
               1='Female';
value yr       1='1997'
               2='1999'
               3='2001';
value ord      0='First'
               1='Recurrent';
value type     0='Non Q-wave'
               1='Q-wave';
value censA    0='Alive'
               1='Dead';
run;

/*  Check Dataset */
proc contents data=temp.whas500 varnum;
ods select attributes position;
run;
```

```
/*****************************************************
 Write basic code and find name of output tables needed
*****************************************************/
ods trace on;
proc npar1way data=temp.whas500 wilcoxon;
class gender;
var hr;
run;
ods trace off;

/*****************************************************
 Output the dataset needed and print to see how it looks
*****************************************************/
proc npar1way data=temp.whas500 wilcoxon;
class gender;
var hr;
ods select KruskalWallisTest;
ods output KruskalWallisTest=myout;
run;

proc print data=myout;
run;

/*****************************************************
 Test Macro variable creation using symput on sample
 output dataset
*****************************************************/
data _null_;
set myout;
if label1="Pr > Chi-Square" then call
     symput('pval',strip(round(nValue1,0.0001)));
run;

%put _user_;
```

```
/*************************************************************
 Create MACRO - used two DO-WHILE loops - first creates
 output in SAS, uses ODS OUTPUT to create a dataset from
 the needed table, finds and stores the needed p-value
 but it does not get put into the PDF file -
 second uses the stored p-values to determine the appropriate
 message and adds to the PDF file (started and closed outside
 of this loop.

 Notice the difference between the difference between the IF
 used for our SYMPUT and that for %IF used for conditional
 execution of the MACRO itself. This is tricky!

 There is a similar distinction between DO and %DO, etc.
*************************************************************/

%macro decisions(mydat, myoutcome, myvars,
myfile="H:\_SAS\MACROS3.pdf");

/* Loop through to get datasets - output not part of PDF results! */
%let i = 1; /* initialize counter */
%let curvar = %scan(&myvars,&i); /* initialize variable */
%do %while(&curvar ne);
     proc npar1way data=&mydat wilcoxon;
     class &curvar;
     var &myoutcome;
     ods select KruskalWallisTest;
     ods output KruskalWallisTest=myout;
     run;
     data _null_;
     set myout;
     if label1="Pr > Chi-Square" then call
          symput("pval&i",strip(round(nValue1,0.0001)));
     run;
     /* P-values are stored by the loop counter &i. This creates
        a macro variable for each value of the loop - PVAL1, PVAL2,
etc.
        When we reference these later we must use multiple & symbols
        when the counter is &i we write &&pval&i to fully evaluate
this
        macro variable */
%let i = %eval(&i +1); /* increase counter */
%let curvar = %scan(&myvars,&i); /* next variable */
%end; /* go back to %do %while until missing */
```

```
/* Create PDF output file with reports */
ods pdf file=&myfile notoc startpage=never;
Ods escapechar='^';
Ods pdf text="^S={font=('Times Roman',14PT,Bold)}OUTCOME VARIABE =
&myoutcome";
%let i = 1;
%let curvar = %scan(&myvars,&i);
%do %while(&curvar ne);
     %put _user_; /* for debugging */
     %if %sysevalf((&&pval&i)>=0 and (&&pval&i) <= 0.05) %then %do;
          Ods pdf text="^S={font=('Times
Roman',14PT,Bold)}Association with &curvar is significant with a p-
value of &&pval&i";
     %end;
     %if %sysevalf((&&pval&i) >  0.05) %then %do;
          Ods pdf text="^S={font=('Times
Roman',14PT,Bold)}Association with &curvar is NOT significant with a
p-value of &&pval&i";
     %end;
%let i = %eval(&i +1);
%let curvar = %scan(&myvars,&i);
%end;
ods pdf close;
%mend;

options nomlogic nomprint;

%decisions(temp.whas500, hr, gender cvd afb sho chf av3 miord
mitype, myfile="H:\_SAS\MACROS3A.pdf");

%decisions(temp.whas500, age, gender cvd afb sho chf av3 miord
mitype, myfile="H:\_SAS\MACROS3B.pdf");
```